



**ESKİŞEHİR OSMANGAZİ ÜNİVERSİTESİ**

Mühendislik Mimarlık Fakültesi

İnşaat Mühendisliği Bölümü

E-Posta: [ogu.ahmet.topcu@gmail.com](mailto:ogu.ahmet.topcu@gmail.com)

Web: <http://mmf2.ogu.edu.tr/atopcu>

# Bilgisayar Destekli Nümerik Analiz

*Ders notları 2014*

**Ahmet TOPÇU**

```
C:\Basic\QBasic.EXE
File Edit View Search Run Debug Calls Utility Options Help
GAUSS.BAS:Gauss
DEFINT I-N
DEFDBL A-H, O-Z
SUB Gauss (n, a(), m, b(), rank, d, k, v)
'-----
' Doğrusal denklem sistemi çözümü (Gauss)
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' a(n,n)*x(n,m)=b(n,m) denklem sisteminden x(n,m) hesaplanır.
' a ve b çağırılan programda depolanmış olmalıdır.
' x(n,m) çözümü b(n,m) matrisinde depolanır.
' n=m ve b(n,n)=1 (birim matris) olarak verilirse
' b(n,n) çözümden sonra a(n,n) nin tersi olur.
' Programın verdiği d, k ve U sayıları yardımıyla
' det a=d*2^k ile hesaplanabilir.
' a tekil ise d=0 döner, çözüm yoktur.
' a nın kondisyon sayısı Kon a=ABS(det a)/U ile
' hesaplanarak matrisin hasta (ill-conditioned) olup
' olmadığını incelenebilir.
'-----
' Machep
Eps = 1
DO
  Eps = Eps / 2
  s = 1 + Eps
LOOP UNTIL s <= 1
Eps = 2 * Eps

' Zero: sıfır sayılacak sayı
rank = 0: v = 1: Zero = 0
FOR i = 1 TO n
  d = 0
  FOR j = 1 TO n
    IF ABS(a(i, j)) > Zero THEN Zero = ABS(a(i, j))
    d = d + a(i, j) * a(i, j)
  NEXT j
  d = SQR(d): v = v * d: 'HADAMARD hacmi
NEXT i
Zero = Zero * Eps

IF n = 1 THEN
' n=1 durumu
IF ABS(a(1, 1)) <= Zero THEN GOTO 100
FOR i = 1 TO m
  b(1, i) = b(1, i) / a(1, 1)
```

# 13

**PROGRAMLAR:** Basit GAUSS indirgeme metodu

**13. PROGRAMLAR: Doğrusal denklem sistemi çözümü- basit GAUSS indirgeme metodu<sup>1</sup>**

$$\begin{bmatrix} a_{11} & a_{12} & \cdot & \cdot & \cdot & a_{1n} \\ a_{21} & a_{22} & \cdot & \cdot & \cdot & a_{2n} \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ a_{n1} & a_{n2} & \cdot & \cdot & \cdot & a_{nn} \end{bmatrix} \begin{bmatrix} x_{11} & x_{12} & \cdot & \cdot & \cdot & x_{1n} \\ x_{21} & x_{22} & \cdot & \cdot & \cdot & x_{2n} \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ x_{n1} & x_{n2} & \cdot & \cdot & \cdot & x_{nn} \end{bmatrix} = \begin{bmatrix} b_{11} & b_{12} & \cdot & \cdot & \cdot & b_{1n} \\ b_{21} & b_{22} & \cdot & \cdot & \cdot & b_{2n} \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ b_{n1} & b_{n2} & \cdot & \cdot & \cdot & b_{nn} \end{bmatrix}$$

$\underline{A} \underline{x} = \underline{b}$  bağıntısında  $\underline{A}$  ve  $\underline{b}$  verildiğinde  $\underline{x}$  bilinmeyenler matrisi,  $\det \underline{A}$ ,  $\underline{A}$  nın kondisyon sayısı ve rankı hesaplanır.  $m=n$  ve karşı taraf matrisi  $\underline{b}$

$$\underline{b} = \underline{I} = \begin{bmatrix} 1 & & & & & \\ & 1 & & & & \\ & & \cdot & & & \\ & & & \cdot & & \\ & & & & \cdot & \\ & & & & & 1 \end{bmatrix}$$

birim matris olarak verilirse  $\underline{A}$  nın ters matrisi,  $\underline{A}^{-1}$  hesaplanır. Hesaplanan  $\underline{x}$  veya  $\underline{A}^{-1}$  matrisi  $\underline{b}$  matrisinde depolanır. Bu nedenle  $\underline{x}$  matrisi için programda boyut açılmamıştır. Hesap sonrası  $\underline{A}$  ve  $\underline{b}$  nin ilk değerleri kaybolur.

**Determinant:**  $\underline{A}$  nın determinanı  $\det \underline{A}$ , önemli bir sabit sayıdır.  $\det \underline{A}=0$  ise çözüm yoktur. Determinant

$$\det \underline{A} = d \cdot 2^k$$

dır. GAUSS programı  $\det \underline{A}$  yı değil  $d$  ve  $k$  sayılarını hesaplar. Bunun nedeni sayı taşmasını önlemektir. Kullanıcı  $\det \underline{A}$  değerini yukarıdaki bağıntıdan hesaplayabilir.

**Kondisyon sayısı(Hadamard):**

$$\text{Kon}(\underline{A})_h = \frac{|\det \underline{A}|}{V}$$

İle tanımlanan bir diğer önemli sayıdır.  $\underline{A}$  matrisinin hasta (ill-condition) bir matris olup olmadığı hakkında bir fikir verir.  $\text{Kon} \underline{A} < 10^{-3}$  durumunda  $\underline{A}$  matrisinin hasta bir matris olduğu, çözümün hatalı olabileceği şüphesi vardır.

**Rank:**  $\underline{A}$  matrisinin lineer bağımsız satır sayısıdır.  $\text{Rank} \underline{A} < n$  ise matris tekildir.  $\det \underline{A} \neq 0$  durumunda daima  $\text{rank} \underline{A} = n$  olur.

**Örnek:**

$$\underline{Ax} = \underline{b} \rightarrow \begin{bmatrix} 6 & -2 & 2 & 4 \\ 12 & -8 & 6 & 10 \\ 3 & -13 & 9 & 3 \\ -6 & 4 & 1 & -18 \end{bmatrix} \begin{bmatrix} x_{11} & x_{12} \\ x_{21} & x_{22} \\ x_{31} & x_{32} \\ x_{41} & x_{42} \end{bmatrix} = \begin{bmatrix} 16 & 1 \\ 26 & -2 \\ -19 & 9 \\ -34 & -4 \end{bmatrix} \rightarrow \underline{x} = ?$$

GAUSS sonucu

```

C:\Basic\QBASIC.EXE
Determinant ve kondisyon hesabı için katsayılar:
Rank= 4 d= .5625 k= 8 U= 45665.9943502821
Denkleminin çözümü (Gauss):
3 .9999999999999999 -2 1
-15.26388888888889 32.95833333333333 48.58333333333333 15.33333333333333
  
```

1. çözüm vektörü

2. çözüm vektörü


<sup>1</sup> Teori ve sayısal örnekler için bak: bölüm 5

**Çözüm:**

$$\underline{x} = \begin{bmatrix} 3 & -15.26 \\ 1 & 32.96 \\ -2 & 48.58 \\ 1 & 15.33 \end{bmatrix}$$

$$\text{Det } \underline{A} = d \cdot 2^k = 0.5625 \cdot 2^8 = 144$$

$$\text{Kon}(\underline{A})_h = |\text{det } \underline{A}| / V = 144 / 45666 = 0.003$$



```

SUB Gauss (n, a(), m, b(), rank, d, k, v)
'-----
' Doğrusal denklem sistemi çözümü (Gauss)
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' a(n,n)*x(n,m)=b(n,m) denklem sisteminden x(n,m) hesaplanır.
' a ve b çağırın programda depolanmış olmalıdır.
' x(n,m) çözümü b(n,m) matrisinde depolanır.
' n=m ve b(n,n)=I (birim matris) olarak verilirse
' b(n,n) çözümden sonra a(n,n) nin tersi olur.
' Programın verdiği d, k ve V sayıları yardımıyla
' det a=d*2^k ile hesaplanabilir.
' a tekil ise d=0 döner, çözüm yoktur.
' a nın kondisyon sayısı Kon a=ABS(det a)/V ile
' hesaplanarak matrisin hasta (ill-conditioned) olup
' olmadığı incelenebilir.
'-----
' Machepe
Eps = 1
DO
  Eps = Eps / 2
  s = 1 + Eps
LOOP UNTIL s <= 1
Eps = 2 * Eps

' Zero: sıfır sayılacak sayı
rank = 0: v = 1: Zero = 0
FOR i = 1 TO n
  d = 0
  FOR j = 1 TO n
    IF ABS(a(i, j)) > Zero THEN Zero = ABS(a(i, j))
    d = d + a(i, j) * a(i, j)
  NEXT j
  d = SQR(d): v = v * d: 'HADAMARD hacmi
NEXT i
Zero = Zero * Eps

IF n = 1 THEN
' n=1 durumu
  IF ABS(a(1, 1)) <= Zero THEN GOTO 100
  FOR i = 1 TO m
    b(1, i) = b(1, i) / a(1, 1)
  NEXT i
  rank = 1: d = a(1, 1): k = 0
  EXIT SUB
END IF

FOR i = 1 TO n - 1
' Kolonda Pivot ara
  T = 0: d = 1: k = i
  FOR j = i TO n
    IF ABS(a(j, i)) > T THEN T = ABS(a(j, i)): k = j
  NEXT j

  IF T <= Zero GOTO 100
  IF k > i THEN
    d = -d: ' Determinantın işaretini deęiş
' i ve k satırlarını deęiş
    FOR j = i TO n
      T = a(i, j)
      a(i, j) = a(k, j)
      a(k, j) = T
    NEXT j
    FOR j = 1 TO m
      T = b(i, j)
      b(i, j) = b(k, j)
      b(k, j) = T
    NEXT j
  END IF

' İndirge
  rank = i
  FOR j = i + 1 TO n
    T = a(j, i) / a(i, i)
    FOR k = i + 1 TO n
      a(j, k) = a(j, k) - T * a(i, k)
    NEXT k
    FOR k = 1 TO m
      b(j, k) = b(j, k) - T * b(i, k)
    NEXT k
  NEXT j
NEXT i
IF ABS(a(n, n)) <= Zero GOTO 100
rank = n

```

**GAUSS alt programı**

```

' Yukarı doğru hesap
FOR j = 1 TO m
  FOR i = n TO 1 STEP -1
    T = 0
    FOR k = i + 1 TO n
      T = T + a(i, k) * b(k, j)
    NEXT k
    b(i, j) = (b(i, j) - T) / a(i, i)
  NEXT i
NEXT j

' Determinat için d ve k yi belirle (WILKINSON)
k = 0
FOR i = 1 TO n
  d = d * a(i, i)
200 k = k + 4
  IF ABS(d) >= 1 THEN d = d * .0625: GOTO 200
300 k = k - 4
  IF ABS(d) < .0625 THEN d = d * 16: GOTO 300
NEXT i
EXIT SUB

100 ' Matris tekil
d = 0: k = 0

END SUB ' Gauss

```